An Introduction to IsoMax

Brad Rodriguez
Forth2020 Virtual Meeting
11 Feb 2023

State Machines

- “A finite-state machine (FSM) ... is an abstract machine that can be in exactly one of a finite number of states at any given time. The FSM can change from one state to another in response to some inputs; the change from one state to another is called a transition. An FSM is defined by a list of its states, its initial state, and the inputs that trigger each transition.” -- Wikipedia

Example

- A very simple stoplight



Forth representation

```
: ew-vehicle-present? ( -- f )
    some-io-address @  1 AND
;
some-io-address CONSTANT ns
some-io-address CONSTANT ew

: red ( addr -- )      4 SWAP ! ;
: yellow ( addr -- )   2 SWAP ! ;
: green ( addr -- )    1 SWAP ! ;

: seconds-delay ( n -- )
    100 * 0 DO  10 MSEC  LOOP
;
```

```
: stoplight
    BEGIN
        ( 1 )  ns green ew red  20 seconds-delay
        ( 2 )  BEGIN ew-vehicle-present? UNTIL
        ( 3 )  ns yellow ew red  5 seconds-delay
        ( 4 )  ns red ew green  20 seconds-delay
        ( 5 )  ns red ew yellow  5 seconds-delay
    AGAIN
;
```

IsoMax representation

```
MACHINE stoplight
ON-MACHINE stoplight
    APPEND-STATE ns-green-1
    APPEND-STATE ns-green-2
    APPEND-STATE ns-yellow
    APPEND-STATE ew-green
    APPEND-STATE ew-yellow

IN-STATE ns-green-1
    CONDITION timer-expired?
    CAUSES     ( no lights change, no timer set )
    THEN-STATE ns-green-2 TO-HAPPEN

IN-STATE ns-green-2
    CONDITION ew-vehicle-present?
    CAUSES  ns yellow  ew red
          5 seconds          ( sets timer to 500 )
    THEN-STATE ns-yellow TO-HAPPEN

IN-STATE ns-yellow
    CONDITION timer-expired?
    CAUSES  ns red  ew green
          20 seconds         ( sets timer to 2000 )
    THEN-STATE ew-green TO-HAPPEN

IN-STATE ew-green
    CONDITION timer-expired?
    CAUSES  ns red  ew yellow
          5 seconds          ( sets timer to 500 )
    THEN-STATE ew-yellow TO-HAPPEN

IN-STATE ew-yellow
    CONDITION timer-expired?
    CAUSES  ns red  ew green
          20 seconds         ( sets timer to 500 )
    THEN-STATE ns-green-1 TO-HAPPEN
```

IsoMax representation, cont'd

```
VARIABLE timer

: timer-expired? ( -- f )
    timer @ if
        -1 timer +!
    then
    timer @ 0=
;

: seconds  100 * timer ! ;


DECIMAL 50000 PERIOD    ( 100 Hz )
0 seconds                    ( sets timer to 0 ticks )
ns-red SET-STATE        ( goes to ns-green-1 )
INSTALL stoplight
```

I/O Trinaries

```
: ew-vehicle-present? ( -- f )
    some-io-address @  1 AND
;
```

```
DEFINE ew-vehicle-present?
    AT-ADDRESS some-io-address
    TEST-MASK 1
    DATA-MASK 0
    FOR-INPUT
```

```
some-io-address CONSTANT ns
some-io-address CONSTANT ew

: red ( addr -- )      4 swap ! ;
```

```
DEFINE ns-red
    SET-MASK 4
    CLR-MASK 3
    AT-ADDRESS some-io-address
    FOR-OUTPUT
```

Questions?

-